
Qui a écrit Les Misérables ?

• Les Misérables : Q180736 

• Auteur : propriété P50

• ?auteur : ce que l’on cherche

 Q180736 P50 ?auteur

https://www.wikidata.org/wiki/Q180736


Qui a écrit Les Misérables ?

SELECT ?auteur 

WHERE {

wd:Q180736 wdt:P50 ?auteur 

}

Réponse : http://tinyurl.com/zyhzwat

wd: → <http://www.wikidata.org/entity/>
wdt: → <http://www.wikidata.org/prop/direct/>
http://tinyurl.com/j3l2mfb

http://tinyurl.com/zyhzwat
http://tinyurl.com/j3l2mfb


Ex 1 : Qui a écrit Candide ?

Réponse : http://tinyurl.com/h4ahyxc

http://tinyurl.com/h4ahyxc


Afficher les libellés

• avec rdfs:label
SELECT ?auteur ?auteurLabel
WHERE {

wd:Q215894 wdt:P50 ?auteur .
?auteur rdfs:label ?auteurLabel .
FILTER (lang(?auteurLabel)="fr") .

}

• avec SERVICE wikibase
SELECT ?auteur ?auteurLabel
WHERE {

wd:Q215894 wdt:P50 ?auteur 
SERVICE wikibase:label {

bd:serviceParam wikibase:language "fr,de,en,th,zh"
} 

}

 http://tinyurl.com/glxx7xh

 http://tinyurl.com/z2v35t5

http://tinyurl.com/glxx7xh
http://tinyurl.com/z2v35t5


Ex 2 : Qui a écrit Vingt mille lieues sous les mers ?

Avec libellé de l’auteur en russe

Réponse 1 : http://tinyurl.com/zv66fzv

Réponse 2 : http://tinyurl.com/hpfzxd5

http://tinyurl.com/zv66fzv
http://tinyurl.com/hpfzxd5


Date de  publication de Vingt mille lieues sous 
les mers

• Spécifier l’année avec la fonction year

SELECT (year(?DatePubli) as ?AnneePubli)

WHERE {
wd:Q183565  wdt:P577 ?DatePubli .

}
 http://tinyurl.com/zo3ynwo

• Rechercher la date de publication

SELECT ?DatePubli

WHERE {
wd:Q183565  wdt:P577 ?DatePubli .

}
 http://tinyurl.com/jyuydm7

http://tinyurl.com/zo3ynwo
http://tinyurl.com/jyuydm7


Ex 3: En quelle année a été peint le tableau La 
Liberté guidant le peuple ?

Réponse  : http://tinyurl.com/jmq8qdb

http://tinyurl.com/jmq8qdb


Œuvres de Johannes Vermeer

• Avec année de création
SELECT ?oeuvre ?oeuvreLabel ?natureLabel (year(?DateCrea) as ?AnneeCrea)

WHERE {

?oeuvre wdt:P170  wd:Q41264 ;

wdt:P31 ?nature ;

wdt:P571 ?DateCrea .

SERVICE wikibase:label { bd:serviceParam wikibase:language "fr" } 

}
 http://tinyurl.com/hjlqme6

• Ordonner les résultats
SELECT ?oeuvre ?oeuvreLabel (year(?DateCrea) as ?AnneeCrea)

WHERE {

?oeuvre wdt:P170  wd:Q41264 ;

wdt:P571 ?DateCrea .

SERVICE wikibase:label { bd:serviceParam wikibase:language "fr" } 

} ORDER BY ?AnneeCrea
 http://tinyurl.com/haqpmaf

http://tinyurl.com/hjlqme6
http://tinyurl.com/haqpmaf


Œuvres de Johannes Vermeer avec options

• Avec année de création et image en option
SELECT ?oeuvre ?oeuvreLabel (year(?DateCrea) as ?AnneeCrea) ?image 

WHERE {

?oeuvre wdt:P170  wd:Q41264 .

OPTIONAL {?oeuvre wdt:P571 ?DateCrea .}

OPTIONAL {?oeuvre wdt:P18 ?image .}

SERVICE wikibase:label { bd:serviceParam wikibase:language "fr" } 

} ORDER BY ?AnneeCrea

 http://tinyurl.com/hxs4gvt

http://tinyurl.com/jdcj3xr


Ex 4: Liste des œuvres de Jules Verne avec 
année de publication

Avec nature de l’œuvre
En option : année de publication, identifiant BnF, image
Tri par nature puis année de publication 

Réponse : http://tinyurl.com/gldeeuh

http://tinyurl.com/gldeeuh


Interroger les qualificatifs

• Ouvrages ayant une édition en français et dont la langue originale n’est pas le français
PREFIX p: <http://www.wikidata.org/prop/>

PREFIX v: <http://www.wikidata.org/prop/statement/>

PREFIX q: <http://www.wikidata.org/prop/qualifier/>

SELECT DISTINCT ?oeuvre ?oeuvreLabel ?format ?URLtxt

WHERE {

?oeuvre wdt:P747  ?edition ;

p:P747 ?DeclarationEdition.

?DeclarationEdition q:P407 wd:Q150.

MINUS { ?item wdt:P364 wd:Q150 }

SERVICE wikibase:label {

bd:serviceParam wikibase:language "fr"

} 

}
 http://tinyurl.com/jjqmlrn

http://tinyurl.com/jjqmlrn


Ex 5 : Liste des œuvres de la série des Voyages 
extraordinaires de Jules Verne

Réponse : http://tinyurl.com/hc3pucq

http://tinyurl.com/hc3pucq


Localiser des éléments

• Fontaines dans Paris
SELECT DISTINCT ?item  ?NomFontaine ?coord

WHERE {

?item wdt:P31 wd:Q483453 .                 

{?item wdt:P131 wd:Q90.}                       

UNION

{?item wdt:P131 ?arr.                       

?arr wdt:P131 wd:Q90. } 

?item rdfs:label ?NomFontaine filter (lang(?NomFontaine) = "fr").  # Titre

?item wdt:P625 ?coord .

}

 http://tinyurl.com/z4wbyfc

http://tinyurl.com/z4wbyfc


Regroupement – Œuvres d’Eugène Delacroix

• Faire des regroupements dans une seule ligne de résultats
SELECT ?oeuvre ?oeuvreLabel (GROUP_CONCAT(year(?DateCrea); separator=" ; ") as ?AnneeCrea)

WHERE {

?oeuvre wdt:P170  wd:Q33477 .

OPTIONAL {?oeuvre wdt:P571 ?DateCrea .}

SERVICE wikibase:label { bd:serviceParam wikibase:language "fr,en" } 

} GROUP BY ?oeuvre ?oeuvreLabel

ORDER BY ?AnneeCrea

 http://tinyurl.com/hcg5son (120 résultats)

 http://tinyurl.com/jha4jwo (125 résultats)

http://tinyurl.com/hcg5son
http://tinyurl.com/jha4jwo


Limiter les résultats

• Personnes nées à Londres
SELECT DISTINCT ?item  ?Nom  

WHERE {

?item wdt:P19 wd:Q84 ;

rdfs:label ?Nom filter (lang(?Nom) = "fr").

} LIMIT 10  http://tinyurl.com/z22dyru

• Nombre de personnes nées à Londres
SELECT (COUNT(DISTINCT ?item) AS ?nombre)

WHERE {

?item wdt:P19 wd:Q84

}  http://tinyurl.com/hj8k9f5

Faire attention à ne pas surcharger l’Endpoint SparQL

http://tinyurl.com/z22dyru
http://tinyurl.com/hj8k9f5


RDF – Ressource Description Framework

https://www.wikidata.org/wiki/Special:EntityData/Q180736.rdf

Les Misérables a pour 
auteur/P50

Victor Hugo

http://www.wikidata.org/entity/Q180736

http://www.wikidata.org/prop/direct/P50

http://www.wikidata.org/entity/Q535

sujet prédicat objet

https://www.wikidata.org/wiki/Special:EntityData/Q180736.rdf


DESCRIBE

 https://query.wikidata.org/#DESCRIBE%20wd%3AQ180736%0A

• Pour obtenir un graphe RDF d’une ressource
DESCRIBE wd:Q180736

https://query.wikidata.org/#DESCRIBE%20wd%3AQ180736%0A


La négociation de contenu

https://www.wikidata.org/entity/Q180736

.html  https://www.wikidata.org/wiki/Q180736

.rdf https://www.wikidata.org/wiki/Special:EntityData/Q180736.rdf

.json https://www.wikidata.org/wiki/Special:EntityData/Q180736.json

.nt (N-Triples) https://www.wikidata.org/wiki/Special:EntityData/Q180736.nt

.n3 (Notation3) https://www.wikidata.org/wiki/Special:EntityData/Q180736.n3

https://www.wikidata.org/wiki/Q180736
https://www.wikidata.org/wiki/Special:EntityData/Q180736.rdf
https://www.wikidata.org/wiki/Special:EntityData/Q180736.json
https://www.wikidata.org/wiki/Special:EntityData/Q180736.nt
https://www.wikidata.org/wiki/Special:EntityData/Q180736.n3


Ressources

 How To SPARQL, University of Manchester and University of Southampton

http://rdf.myexperiment.org/howtosparql

 Exemple de requêtes avec l’Enpoint SparQL de Wikidata, Mediawki

https://www.mediawiki.org/wiki/Wikibase/Indexing/SPARQL_Query_Examples

 Wikidata Enpoint SparQL et les peintures de Goya, par Shonagon, 10/2015

http://zone47.com/dozo/wikidata-enpoint-sparql-et-les-peintures-de-goya

 Mais combien y a-t-il de femmes scientifiques sur Wikipédia ?, par Ash Crow, 10/2015 

http://blog.ash.bzh/mais-combien-y-a-t-il-de-femmes-scientifiques-sur-wikipedia-this-time-with-sparql/

 Wikidata, SPARQL et chiens de traîneaux, par Ash Crow, 10/2015 

http://blog.ash.bzh/wikidata-sparql-et-chiens-de-traineaux/

 Wikidata, SPARQL et dynasties électives, par Ash Crow, 10/2015

http://blog.ash.bzh/wikidata-sparql-et-dynasties-electives/

http://rdf.myexperiment.org/howtosparql
https://www.mediawiki.org/wiki/Wikibase/Indexing/SPARQL_Query_Examples
http://zone47.com/dozo/wikidata-enpoint-sparql-et-les-peintures-de-goya
http://blog.ash.bzh/mais-combien-y-a-t-il-de-femmes-scientifiques-sur-wikipedia-this-time-with-sparql/
http://blog.ash.bzh/wikidata-sparql-et-chiens-de-traineaux/
http://blog.ash.bzh/wikidata-sparql-et-dynasties-electives/

